

## Analyse, simulations d'une alimentation régulée

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet : [page d'accueil](#)

d'autres pdf, sur différents sujets : [liste des PDE](#)

Ce document montre l'étude et l'analyse d'une alimentation stabilisée. Une approche par les systèmes bouclés explique, sur un exemple concret, le risque d'instabilité, selon le choix du condensateur connecté en sortie. Ce problème, connu, est commenté dans les data sheet des fabricants de régulateurs intégrés.

### 1) Introduction, étude simplifiée

A partir d'une tension continue  $V_{IN}$  disponible, on cherche à fabriquer une tension continue, constante, indépendante du courant débité. La structure série est une réponse, où un élément ballast délivre une tension  $V_{OUT}$ , qui alimente l'utilisateur (charge symbolisée par  $R_L$ ).

Le principe de ce régulateur linéaire est de comparer une tension de référence constante  $V_C$  (issue d'un montage électronique dédié, mais prévu pour une faible puissance), à une image de la tension  $V_{OUT}$ .

Si  $V_{OUT}$  vient à diminuer, l'écart  $k V_{OUT}$  vis-à-vis de  $V_C$  augmente, et cette évolution commande le ballast qui réduit la chute de tension entre  $V_{IN}$  et  $V_{OUT}$ , ce qui maintient  $V_{OUT}$  à sa valeur.

Pour un récepteur consommant un courant important, le ballast est formé d'un transistor de puissance. Nous avons choisi un PMOS : sa commande est en Grille, sa sortie est en Drain : il est inverseur. Il faut donc permuter les entrées + - de l'amplificateur différentiel A. On pose A, le potentiel de sortie de l'amplificateur.  $V_A = V_G$ . L'image de la tension est prise au travers un pont diviseur de tension :  $k = R_2/(R_1+R_2)$ . Le courant dans le pont est négligeable devant celui circulant dans  $R_L$ .




Schéma de principe




Schéma réel

#### Polarisation

On suppose  $e^+ = e^-$  de l'A. Op. C'est à dire :  $V_C = k V_{OUT}$ , d'où  $V_{OUT} = V_C / k$ . La précision sur la valeur numérique de  $V_{OUT}$  est directement celle de  $V_C$ .

Connaissant  $R_L$  on a  $I_D$ . On suppose a priori  $|V_{GS}| > |V_T|$  pour avoir conduction, et  $|V_{DS}| > |V_{GS}| - |V_T|$  pour avoir le mode saturé. Cela signifie également  $|V_{DS}| > |V_{DSsat}|$ , c'est à dire  $V_{IN} - V_{OUT} > |V_{DSsat}|$ . Cela forme une condition sur les tensions pour que l'alimentation régule avec le transistor agissant en régime saturé, c'est-à-dire en mode source de courant contrôlée par une tension  $I_D = f(V_{GS})$ .

Dans ces conditions, on a, en première approximation :  $I_D = \beta (|V_{GS}| - |V_T|)^2$  avec  $\beta$ , constante du transistor, connue, et  $I_D$  circulant de la Source vers le Drain dans le transistor. On peut alors en déduire  $V_{GS}$ , et par suite  $V_A$ .

#### Etude en variation, petits signaux, linéarisé

On a, en mode source de courant, la relation  $i_D = g_m v_{GS}$  (formalisme des lettres minuscules pour les variations).

Pour tenir compte qu'à  $v_S$  fixé, le courant diminue quand  $v_G$  augmente, on affecte un signe « - » le coefficient  $g_m$ .

Faisons le bilan des équations :

$$v_A = A (e^+ - e^-) = A (k V_{OUT} - V_C) \quad v_{GS} = V_G - v_S = v_A - V_{IN}$$

$$V_{OUT} = R_L i_D$$




Schéma fonctionnel résultant

En petits signaux, système linéarisé, on peut appliquer le théorème de superposition :  $v_{OUT}$  est formé de la contribution de  $v_C$ , à  $v_{IN} = 0$ , plus la contribution de  $v_{IN}$ , à  $v_C = 0$ . On prend comme hypothèse que la source  $V_{IN}$  n'a pas de variation dynamique, soit  $v_{IN} = 0$ , ce qui simplifie le schéma fonctionnel. De plus, on retire le signe « - » dans la chaîne directe en permutant les signes des entrées du comparateur.

Il en résulte un schéma de travail simplifié.

L'expression de la TBF donne :

$$\frac{v_{OUT}}{v_C} = \frac{A g_m R_L}{1 + A g_m R_L k} = \frac{1}{k} \text{ car } A g_m R_L k \gg 1.$$




Schéma fonctionnel simplifié

## 2) Etude plus réaliste sur un exemple concret, prédétermination de la stabilité

Nous allons montrer, par une approche simplifiée, que le montage peut être instable.

a) L'amplificateur A est un ua741 (issu de eval.lib). On adopte le modèle harmonique suivant :

- un plateau à 106 dB,
- un passage à  $-20$  dB/déc à  $f_1 = 5$  Hz,
- un passage à  $-40$  dB/déc à  $f_2 = 1$  MHz.



Il vient l'expression de  $A(p)$ , transmittance de Laplace de l'A.Op. seul :

$$A(p) = \frac{A_0}{(1 + \tau_1 p)(1 + \tau_2 p)}$$

où :

- $A_0 \approx 200\,000$ ,
  - $1/(2\pi\tau_1) = 5$  Hz
  - $1/(2\pi\tau_2) = 1$  MHz.
- La deuxième cassure est à 0 dB.

Il présente une résistance interne que l'on modélisera par  $R_a = 25 \Omega$  :




b) Le PMOS est un IRF9140 (issu de eval.lib).

C'est un transistor de puissance, en boîtier TO3.


Il faut tenir compte des capacités  $C_{GS}$  et  $C_{GD}$ . La valeur de  $g_m$  est liée à la polarisation du transistor. On va admettre que  $g_m = 6,22$  A/V et  $C_{GS} = 875$  pF, ce que nous vérifierons ultérieurement.

Pour simplifier cette pré-étude, on va négliger la contribution de  $C_{GD}$ , placée entre une source de tension quasi parfaite et  $v_{OUT}$ , proche de 0 V en dynamique, par  $R_L$ .




Le schéma électrique correspondant, est, en dynamique ( $v_{IN}$  est à la masse) montré ci-contre.

Entre  $v_A$  et  $v_G$ , il apparaît un pont diviseur de tension, de fonction de transfert  $\frac{1}{1+\tau p}$  avec  $\tau = R_a C_{GS}$ .



Le schéma fonctionnel en est :

$$\tau = 25 \Omega \times 875 \text{ pF} \approx 22 \text{ ns}$$



**Schéma fonctionnel plus complet**

Cette analyse simplifiée montre que le système bouclé est de type passe bas avec gain, d'ordre 3. Une étude harmonique permet de vérifier l'instabilité du circuit :

On a :

$$TBO(p) = \frac{A_0 g_m R_L k}{(1 + \tau_1 p)(1 + \tau_2 p)(1 + \tau p)}$$


Application numérique :

On fixe  $R_L = 5 \Omega$ .

$$A_0 g_m R_L k = 200\,000 \times 6,22 \times 5 \times 0,5 = 3,11 \cdot 10^6, \text{ soit } 130 \text{ dB}$$

$$1/(2\pi\tau) = 1/(2\pi \times 22 \text{ n}) = 7,275 \text{ MHz.}$$

La réponse harmonique asymptotique est :



Avec cette modélisation des circuits relativement simple, on diagnostique que le système est **instable**, car la marge de phase est négative (repérée en jaune, lors de la traversée à 0 dB de la courbe de gain).

### 3) Vérification de l'instabilité

On choisit de placer 5 V constant en entrée de consigne. Idéalement, la sortie doit être à 10 V.

La **directive .OP** permet, pour 5 V à l'entrée  $V_C$ , de vérifier les tensions continues :

```
NODE  VOLTAGE
( C )  5.0000
( A )  10.5440
( OUT ) 9.9999
( k )  4.9999
```

Ainsi que la polarisation du transistor :

```
MODEL  IRF9140
ID      -2.01E+00
VGS     -4.46E+00
VDS     -5.00E+00
VBS     0.00E+00   GM  6.22E+00
VTH     -3.67E+00
VDSAT  -6.45E-01  CGS 8.75E-10
```

alimentation stabilisée

\*fichier alim.cir

.lib eval.lib ; contient LF411 et IRF9140

Valim 15 0 DC 15 ; alim puissance entrée

Ve C 0 PULSE (5 5.1 25u 10n 10n 25u 50u) DC=5

X1 k C 15 0 A ua741 ; e+ e- V+ V- Sortie

MP OUT A 15 15 IRF9140 ; PMOS D G S Sub

RL OUT 0 5 ; charge nominale

R1 OUT k 1k ; pont

R2 k 0 1k ; diviseur

.OP

.TRAN 10n 80u 0u 10n

.probe

.end

**Netlist, prête à simuler**


La sortie est quasiment à 10 V, comme attendu. Avec  $R_L = 5 \Omega$  le courant est **2 A**.

On vérifie, qu'à ce point de polarisation,  $g_m = 6,22 \text{ A/V}$  et  $C_{GS} = 875 \text{ pF}$ , comme fixé au préalable.

La valeur de  $\beta = \frac{K_P}{2} \frac{W}{L} = \frac{10,15 \mu}{2} \frac{1,9}{2 \mu} = 4,82$  permet de vérifier la polarisation prédéterminée :

En tenant compte de la résistance d'accès à la source, donnée dans le modèle de l'IRF9140,  $= 70,6 \text{ m}\Omega$ , soit une chute de tension de 141,2 mV, et par suite  $|V_{GS}|$  intrinsèque  $= 4,46 - 0,141 = 4,319 \text{ V}$ , on retrouve :  $I_D = \beta (|V_{GS}| - |V_T|)^2 = 4,82 (4,319 - 3,67)^2 = 2 \text{ A}$ .

Lançons une **analyse transitoire**, en simulant un petit écart de tension appliqué à la consigne :  $V_C(t)$  passe de 5 V à 5,1 V. On peut raisonnablement admettre que la polarisation n'est que peu modifiée, et que les valeurs numériques restent ainsi conservées. Le changement de consigne se fait toutes les 25  $\mu\text{s}$ .



La simulation montre que la tension de sortie est bien à 10 V dès  $t = 0$ . Mais une oscillation démarre, bien visible à notre échelle vers 15  $\mu\text{s}$ . Les différentes non linéarités (caractéristiques du transistor, de l'A. Op.), font que l'amplitude de l'oscillation reste bornée. Sa fréquence est 530 kHz.

A  $t = 25 \mu\text{s}$ , la sortie s'oriente vers 10,2 V, en obéissant au changement de consigne, mais le système continue d'osciller. A  $t = 50 \mu\text{s}$ , la sortie retourne vers 10 V, toujours instable.


On peut dire que le système répond à l'entrée, mais oscille en permanence.

Cette simulation exploite des modèles électroniques des composants, bien plus complets que ceux utilisés pour la prédétermination. Par exemple, la réponse de l'A.Op. est plus complexe (courbe de phase qui tourne davantage au voisinage de 1 MHz, slew rate),  $C_{GD}$  intervient, etc, mais la conclusion établie avec des hypothèses simplificatrices est conservée : le système est instable.

Nous allons voir maintenant, comment, en pratique, pourquoi ce type d'alimentation fonctionne sans osciller.

### 4) Schéma plus complet

Nous allons modifier le schéma pour tenir compte maintenant d'un condensateur connecté sur la sortie. La charge n'est plus une simple résistance  $R_L$ , mais une association  $R_L // C_L$ . Ce condensateur  $C_L$  est souvent du type polarisé de forte valeur. Sa technologie est telle qu'il présente, en interne, une résistance série  $R_S$  (notée usuellement ESR). La charge est donc  $R_L // (C_L \text{ série } R_S)$ . En toute rigueur, il présente également une inductance série, que nous négligerons.



Nous avons une impédance de charge  $R_L // (C_L \text{ série } R_S)$  :  $Z(p) = \frac{R_L(1 + R_S C_L p)}{1 + (R_S + R_L)C_L p}$

Le nouveau schéma fonctionnel montre une TBO (p) :

$$\frac{A_0 \ g_m \ R_L \ k}{(1 + \tau_1 p)(1 + \tau_2 p)(1 + \tau p)} \frac{1 + R_S C_L p}{1 + (R_S + R_L) C_L p}$$




Schéma fonctionnel plus complet

Sa réponse harmonique reprend la précédente, mais avec un pôle supplémentaire à  $1/[2\pi(R_S+R_L)C_L]$  et un zéro à  $1/(2\pi R_S C_L)$ .


Application numérique :  $1/(2\pi\tau) = 7,275 \text{ MHz}$        $1/(2\pi\tau_1) = 5 \text{ Hz}$        $1/(2\pi\tau_2) = 1 \text{ MHz}$ .

Prenons  $C_L = 100 \mu\text{F}$ , et  $0,2 \ \Omega$  la résistance série de ce condensateur :

Le pôle supplémentaire est à  $1/[2\pi(R_S+R_L)C_L] = 1/(2\pi \cdot 5,2 \cdot 100\mu) = 306 \text{ Hz}$

Le zéro est placé à  $1/(2\pi R_S C_L) = 1/(2\pi \cdot 0,2 \cdot 100\mu) = 8 \text{ kHz}$ .

La réponse harmonique asymptotique est :



Le zéro agit en avance de phase : à 8 kHz, la phase remonte, et à la traversée de 0 dB, elle est bien au-dessus de -180 °. Le système est stabilisé. Une lecture sur les asymptotes obliques permet d'évaluer une marge supérieure à 45 °.

Remarques :


- Il est curieux de constater que c'est un défaut du condensateur (la résistance série) qui permet à l'alimentation de ne pas osciller ! Ou, dit autrement, si on place un condensateur parfait, le système oscille, car la TBO est d'ordre 4 passe bas : le zéro  $1/(2\pi R_S C_L)$  est décalé vers l'infini. On peut donc être amené à placer une résistance supplémentaire, pour créer le zéro stabilisateur.
- Mais par contre, si la résistance série est trop importante, le zéro  $1/(2\pi R_S C_L)$  est décalé vers la gauche, et la marge de phase est insuffisante.

Il faut donc un compromis résistance série / condensateur de sortie pour assurer le bon fonctionnement de l'alimentation. C'est le renseignement que donnent les constructeurs de régulateur intégré, comme Texas Instrument, avec le TPS76425.

Sa structure interne est celle de notre schéma d'étude : PMOS, amplificateur d'erreur, pont diviseur sur la sortie et référence d'entrée.

La data sheet indique clairement, pour un condensateur de sortie de 10 µF, que la résistance série (intrinsèque ou externe) ne doit pas être inférieure à 0,3 Ω ni supérieure à 9 Ω.

A noter que le TPS76425 est un « LDO » (low drop out), c'est-à-dire qui présente une très faible chute de tension entre Drain et Source, ce qui n'est pas le cas de notre schéma.




Simulation :

Ajoutons le condensateur de 100 µF en série avec 0,2 Ω, et relançons l'analyse transitoire.

```

Cch OUT 10 100u
resr 10 0 0.2
modification de la netlist
    
```




On a clairement un système stable : la tension de sortie passe de 10 V à 10,2 V (quand la consigne passe de 5 V à 5,1 V) avec une réponse oscillatoire qui montre un faible dépassement et un amortissement rapide : il faut environ 2 µs pour que la tension de sortie se stabilise après un changement instantané de tension de consigne.

## 5) Schéma final

Dans ce type d'alimentation, il faut prévoir une protection en courant.

Pour ce faire, on surveille le courant traversant le ballast.

Une résistance en série permet une conversion courant → tension, et un montage électronique bloque le ballast quand une valeur seuil de tension est atteinte. Le schéma résultant en est :




Le principe de la protection est : si  $R \times I$  dépasse 0,6 V environ, le transistor bipolaire (PNP 2N2907) se met à conduire. La tension  $|V_{CE}|$  descend (elle ira à  $V_{CEsat}$  avec  $I_B \gg I_C/\beta$ ), réduisant ainsi  $|V_{GS}|$ , ce qui a pour conséquence de réduire  $I_D$ , et donc le courant allant vers  $R_L$ . La résistance  $r$  limite néanmoins cette action, et évite également de faire entrer un courant trop important dans l'A. Op.

Avec  $R = 0,22 \Omega$ ,  $r = 150 \Omega$ , cet étage de protection agit à  $I_D = 5 A$ .

Pour vérifier cet étage de protection, on reproduit ce qu'on ferait sur table : on fait varier  $R_L$  de l'infini vers  $0 \Omega$ .

On exploite le composant « Voltage-Controlled Switch », qui est une résistance commandée en tension. Sa valeur est entre  $R_{ON}$  et  $R_{OFF}$ , respectivement quand on applique  $V_{ON}$   $V_{OFF}$  sur une tension de commande.



- En utilisation « résistance variable », on applique une tension de commande qui évolue progressivement entre  $V_{OFF}$  (d'où résistance très grande) à  $V_{ON}$  (d'où court-circuit).

- En utilisation « interrupteur », on applique  $V_{ON}$  et  $V_{OFF}$  en tout ou rien.

La loi de commande pour passer de  $R_{ON}$  à  $R_{OFF}$  est complexe, mais monotone.

Pour tester l'alimentation, on fixe  $V_C$  à 5 V, et on fait varier le courant débité.

### Test statique

Idéalement, la sortie doit être à 10 V.

Le Switch est utilisé en résistance variable, allant de  $10 k\Omega$  à  $1 m\Omega$ .

On observe la tenue de la tension de sortie face à cette montée en courant.

L'analyse est .DC, c'est-à-dire en continu.


La protection en courant est en service.

```

alimentation stabilisée
*fichier test_alim.cir
.lib eval.lib
Valim 150 0 DC 15 ; PULSE (15 16 25u 10n 10n 25u 50u)
R 150 15 0.22 ; résistance de protection 0.22 pour 5 A
Q1 AA 15 150 Q2N2907A ; CBE
RA AA A 150
Ve C 0 DC=5
X1 k C 15 0 A ua741 ; e+ e- V+ V- Sortie
MP OUT A 15 15 IRF9140 ; PMOS D G S Sub
Vcom com 0 PULSE (2.26 2.49 25u 10n 10n 25u 50u)
.Model int Vswitch (Ron=0.001 Roff=10k Von=5 Voff=0)
Sint OUT 0 com 0 int
R1 OUT k 1k
R2 k 0 1k
Cch OUT 10 100u
resr 10 0 0.2
.OP
.DC Vcom 0 5 0.01
.TRAN 10n 80u 0u 10n
.probe
.end

```

**Netlist, prête à simuler**




La tension de sortie est à 10 V tant que le courant n’atteint pas 5 A. Puis, quand la résistance de charge est inférieure à 2 Ω, pour tendre vers un court-circuit, le courant est maintenu à 5 A, et la tension obéit à la loi d’Ohm, et non plus à l’asservissement :  $V_{OUT}$  tend vers 0 V.

### Test dynamique


L’analyse est .TRAN.

Le Switch est utilisé en interrupteur. On choisit des valeurs à la tension de commande du Switch de façon à ce, par loi d’Ohm, le courant débité soit de 1 A à 3 A, c’est-à-dire  $R_L$  qui passe de 10 Ω à 3,33 Ω, instantanément.



On observe la tenue de la tension de sortie face à ce brusque appel de courant.

Bien que câblée, la protection au courant n’intervient pas, car le courant est inférieur à 5 A.



On voit très clairement une brève chute de tension (environ 0,4 V) lors de l’appel de courant, puis l’action de l’asservissement qui rétablit à 10 V la tension de sortie, et ce, en 3 μs environ.

Lors de la transition inverse, un pic de tension (de 0,4 V) se présente au-delà de 10 V, puis le rétablissement à 10 V se fait en 3 μs également.

### Conclusion

Notre alimentation est un régulateur linéaire, en structure série. Elle permet d’assurer 10 V en sortie (avec une tension de consigne de 5 V, qui peut être issue d’une diode zener), et ce, tant que le courant débité ne dépasse pas la sécurité fixée à 5 A. Son temps de retour à la valeur nominale de tension de sortie lors d’un appel de courant est de 3 μs.

Cette étude a montré que le choix d’avoir une boucle à fort gain, par la présence de l’A.Op., entraîne un système bouclé instable. On a vérifié que de placer un condensateur de sortie, équipé d’une résistance série, stabilisait le montage.

articles 1 à 43 : sur le livre

**Tableau récapitulatif des articles PDF disponibles sur ce site**

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	<a href="#">présentation</a>	<a href="#">document PDF</a>
44	Exemples basiques et des exercices...	<a href="#">présentation</a>	<a href="#">document PDF</a>
45	Un exemple de circuit passif	<a href="#">présentation</a>	<a href="#">document PDF</a>
46	Un oscillateur Colpitts	<a href="#">présentation</a>	<a href="#">document PDF</a>
47	Compensation en fréquence des amplificateurs opérationnels	<a href="#">présentation</a>	<a href="#">document PDF</a>
48	Un amplificateur à transistors bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
49	Une bascule D Flip Flop CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
50	Une porte XOR à transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
51	Un VCO à 12 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
52	Une PLL à moins de 20 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
53	Un oscillateur à résistance négative	<a href="#">présentation</a>	<a href="#">document PDF</a>
54	Une charge électronique	<a href="#">présentation</a>	<a href="#">document PDF</a>
55	Un amplificateur en classe C	<a href="#">présentation</a>	<a href="#">document PDF</a>
56	Le monostable 74 123	<a href="#">présentation</a>	<a href="#">document PDF</a>
57	Un amplificateur en classe D	<a href="#">présentation</a>	<a href="#">document PDF</a>
58	Le transformateur en linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
59	La loi d'ohm thermique	<a href="#">présentation</a>	<a href="#">document PDF</a>
60	Le transformateur en non linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
61	Robustesse d'un oscillateur en anneau	<a href="#">présentation</a>	<a href="#">document PDF</a>
62	Une alimentation stabilisée	<a href="#">présentation</a>	<a href="#">document PDF</a>
63	Modélisation d'un haut-parleur	<a href="#">présentation</a>	<a href="#">document PDF</a>
64	Un synthétiseur de fréquence	<a href="#">présentation</a>	<a href="#">document PDF</a>
65	Un ampli audio de Sparkfun	<a href="#">présentation</a>	<a href="#">document PDF</a>
66	Simulation logique et analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
67	Un oscillateur à relaxation	<a href="#">présentation</a>	<a href="#">document PDF</a>
68	Lecteur de TAG RFID 125 kHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
69	Diagramme de l'œil avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
70	Un amplificateur hautes fréquences	<a href="#">présentation</a>	<a href="#">document PDF</a>
71	Une bizarrerie enfin expliquée...	<a href="#">présentation</a>	<a href="#">document PDF</a>
72	Comprendre le paramétrage de la FFT	<a href="#">présentation</a>	<a href="#">document PDF</a>
73	La relation de Bennett	<a href="#">présentation</a>	<a href="#">document PDF</a>
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	<a href="#">présentation</a>	<a href="#">document PDF</a>
75	Une horloge biphase sans recouvrement	<a href="#">présentation</a>	<a href="#">document PDF</a>
76	Quelques simulations sur la diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
77	Un ampli classe A, avec transformateur de sortie	<a href="#">présentation</a>	<a href="#">document PDF</a>
78	Des stimuli pour PSpice	<a href="#">présentation</a>	<a href="#">document PDF</a>
79	Simuler le TL431 : zener ajustable	<a href="#">présentation</a>	<a href="#">document PDF</a>
80	Un ADC flash	<a href="#">présentation</a>	<a href="#">document PDF</a>
81	Une chaîne d'acquisition : S&H, ADC, DAC	<a href="#">présentation</a>	<a href="#">document PDF</a>
82	Un amplificateur 50 MHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
83	Un dérivateur non inverseur	<a href="#">présentation</a>	<a href="#">document PDF</a>
84	Un amplificateur bipolaire avec push pull CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
85	Rôle des répéteurs logiques dans un circuit intégré	<a href="#">présentation</a>	<a href="#">document PDF</a>
86	Un driver logique CMOS pour charge 50 ohms	<a href="#">présentation</a>	<a href="#">document PDF</a>
87	Des triggers de Schmitt et des applications	<a href="#">présentation</a>	<a href="#">document PDF</a>
88	Un filtre gaussien analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
89	Un générateur de bruit rose	<a href="#">présentation</a>	<a href="#">document PDF</a>

90	Un anémomètre à fil chaud : simulation comportementale	<a href="#">présentation</a>	<a href="#">document PDF</a>
91	Un oscillateur à pont de Wien stabilisé par CTN	<a href="#">présentation</a>	<a href="#">document PDF</a>
92	L'emballement thermique d'une diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
93	Les puissances dans un amplificateur	<a href="#">présentation</a>	<a href="#">document PDF</a>
94	Asservissement de puissance dans une résistance	<a href="#">présentation</a>	<a href="#">document PDF</a>
95	Asservissement de la puissance émise par une antenne radio	<a href="#">présentation</a>	<a href="#">document PDF</a>
96	Un driver de LED de puissance	<a href="#">présentation</a>	<a href="#">document PDF</a>
97	Exploiter Pspice pour simuler des filtres numériques	<a href="#">présentation</a>	<a href="#">document PDF</a>
98	Un filtre en cosinus surélevé avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
99	Effet de la température sur un amplificateur en classe A	<a href="#">présentation</a>	<a href="#">document PDF</a>
100	Un amplificateur à transistors JFET et bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
<b>Supplément, hors article :</b>			
<b>mon cours « Electronique pour les communications numériques », <a href="#">polycopié couleur 201 pages en pdf</a></b>			

[retour à l'écran d'accueil de ce site](#)